


Cucarachas cósmicas

12



Clasificación:

JUEGO DE EVASION

Utiliza las teclas mostradas a continuación para ayudar a Harry el Hambriento, el hombre peludo de las montañas, a abandonar su campamento situado en la parte inferior de la pantalla, y maniobrar a través de las peligrosas cucarachas cósmicas para robar la comida que se encuentra en la parte superior de la pantalla.

Harry debe entonces volver a su campamento y repetir el proceso tantas veces como sea posible.

Para mover utiliza: "A" arriba, "Z" abajo, "coma" izquierda y "punto" derecha.

SUGERENCIAS DE PROGRAMACION

Este juego puede modificarse para hacer que la cucaracha más cercana se mueva hacia Harry más frecuentemente o para cambiar la velocidad de las cucarachas y la velocidad de Harry.

Variables del programa

DIR(5) Direcciones de cada cucaracha.
CC Cucaracha más cercana (1-5).

HCOL, HFIL	Columna y fila de Harry.
HVLX, HVLY	Velocidad de Harry en cada dirección.
CUCAS\$(5)	Carácter de cada cucaracha.
BL\$	Cadena usada para borra un carácter.
CUCAX(5), CUCA Y(5)	Coordenadas de las cucarachas.
NT	Número de viajes terminados.
CARGADO	¿Está Harry con comida?
CT	Número de cucarachas en acción.
I	Temporal.

Estructura del programa

<i>Líneas</i>	<i>Función/Actividad</i>
10-130	Inicializar
140-300	Bucle principal
310-360	Harry arroja la comida
370-470	Dibujo de la comida/campamento
480-510	Dibujo de las cucarachas
520-610	Movimientos de las cucarachas
620-670	Lectura del teclado
680-750	Fin del juego

```

Inicializar juego

10 MODE 1:INK 0,13:INK 1,0:INK 2,3:INK 3
  ,19:PAPER 0:BORDER 13:CLS
20 SYMBOL 240,3,0,240,8,4,7,15,31:SYMBOL
  241,31,15,15,4,8,240,0,3:SYMBOL 242,136
  ,136,136,144,166,255,255,246:SYMBOL 243,
  246,255,255,166,144,136,136,136
30 SYMBOL 244,68,34,17,17,17,63,127,255:
  SYMBOL 245,255,127,63,17,17,17,34,68:SYM
  BOL 246,64,32,16,17,17,250,252,252:SYMBOL
  L 247,252,252,250,17,17,16,32,64
40 CUCA$(1)=CHR$(244)+CHR$(246)+CHR$(10)
  +CHR$(8)+CHR$(8)+CHR$(245)+CHR$(247)
50 FOR I=2 TO 5:CUCA$(I)=CHR$(240)+CHR$(
  242)+CHR$(10)+CHR$(8)+CHR$(8)+CHR$(241)+
  CHR$(243):NEXT I
60 CC=5:HCOL=20:HFIL=24:HVLX=0:HVLY=-1:C

```

```
COL(1)=3:FOR I=2 TO 5:CCOL(I)=2:NEXT
70 DIR(1)=-3:DIR(2)=2:DIR(3)=-2:DIR(4)=1
:DIR(5)=-1
80 BL$="┘┘"+CHR$(10)+CHR$(8)+CHR$(8)+"┘┘
"
90 CUCAX(1)=34:CUCAY(1)=3:CUCAX(2)=2:CUC
AY(2)=7:CUCAX(3)=32:CUCAY(3)=11:CUCAX(4)
=3:CUCAY(4)=15:CUCAX(5)=33:CUCAY(5)=19
100 NT=-1:GOSUB 310:CARGADO=1:GOSUB 370:
CARGADO=0
110 EVERY 30,1 GOSUB 620
120 FOR CT=1 TO 5:GOSUB 480:NEXT
```

Bucle principal del programa

```
130 :
140 FOR CT=1 TO 5
150 GOSUB 520
160 NEXT
170 LOCATE HCOL,HFIL:PRINT"┘"
180 HCOL=HCOL+HVLX:HFIL=HFIL+HVLY:CC=INT
(ABS((HFIL-2)/4))+1
190 CC=CC+(CC=6)
200 IF HCOL<1 THEN HCOL=35
210 IF HCOL>35 THEN HCOL=1
220 IF HFIL<2 THEN HFIL=2
230 IF HFIL>24 THEN HFIL=24
240 PEN 1:LOCATE HCOL,HFIL:PRINT CHR$(24
8)
250 T1=CUCAY(CC)-HFIL:T2=CUCAX(CC)-HCOL
260 IF(T1=-1 OR T1=0)AND(T2=-1 OR T2=0)T
HEN 680
270 IF(CARGADO=0)AND HCOL=21 AND HFIL=2
THEN GOSUB 370:CARGADO=1:PRINT CHR$(7)
280 IF CARGADO AND(HCOL=20)AND(HFIL=24)T
HEN GOSUB 310
290 GOTO 140
```

Harry arroja la comida

```
300 :
310 GOSUB 370
```

```
320 CARGADO=0:NT=NT+1
330 LOCATE 1,1:PEN 1:PRINT"VIAJES:␣";NT
340 FOR I=1 TO 1000 STEP 20:SOUND 1,I,2,
15,1,1,0:NEXT
350 RETURN
```

Dibujo de la comida y campamento

```
360 :
370 IF CARGADO THEN T=1 ELSE T=0
380 MOVE 320,390:DRAWR 0,-16,T
390 DRAWR 20,0,T:DRAWR 0,16,T
400 DRAWR-20,0,T:DRAWR 20,-16,T
410 MOVER 0,16:DRAWR-20,-16,T
420 MOVER 4,16:DRAWR 0,6,T
430 DRAWR 12,0,T:DRAWR 12,-4,T
440 MOVE 308,4:DRAWR 12,0,1
450 DRAWR-6,6,1:DRAWR-6,-6,1
460 RETURN
```

Dibujo de las cucarachas

```
470 :
480 LOCATE CUCAX(CT),CUCAY(CT)
490 PEN CCOL(CT):PRINT CUCA$(CT)
500 RETURN
```

Movimientos de las cucarachas

```
510 :
520 LOCATE CUCAX(CT),CUCAY(CT):PRINT BL$
:CUCAX(CT)=CUCAX(CT)+DIR(CT)
530 IF CUCAX(CT)<=1 THEN CUCAX(CT)=1:DIR
(CT)=-DIR(CT)
540 IF CUCAX(CT)>=35 THEN CUCAX(CT)=34:D
IR(CT)=-DIR(CT)
550 GOSUB 480
560 IF CC<>CT THEN RETURN
570 IF RND(1)<0.4-NT/10 THEN RETURN
```

```

580 IF CUCAX(CT)>HCOL THEN DIR(CT)=-ABS(
DIR(CT)):RETURN
590 DIR(CT)=ABS(DIR(CT))
600 RETURN

```

Lectura teclado

```

610 :
620 IF INKEY(39)>=0 THEN HVLX=-1:HVLY=0:
RETURN
630 IF INKEY(31)>=0 THEN HVLX=1:HVLY=0:R
ETURN
640 IF INKEY(69)>=0 THEN HVLY=-1:HVLX=0:
RETURN
650 IF INKEY(71)>=0 THEN HVLY=1:HVLX=0:R
ETURN
660 RETURN

```

Fin del juego

```

670 :
680 LOCATE 5,24:PEN 1:PRINT"POBRE_HARRY"
690 ENT 1,45,21,11,10,-1,1,10,1,1,5,-1,1
:FOR CT=1 TO 5:GOSUB 480:NEXT
700 DEG:FOR I=0 TO 180 STEP 6:INK 2,RND(
1)*26:INK 3,RND(1)*26
710 SOUND 1,1000*SIN(I),15,15,0,1
720 NEXT
730 LOCATE 5,24:PEN 1:PRINT"PULSA_UNA_TE
CLA_PARA_OTRO_JUEGO"
740 IF INKEY$<>""THEN 740
750 IF INKEY$=""THEN 750 ELSE RUN

```

Tablas de comprobación

10 =	1700	60 =	6575	110 =	570
20 =	7578	70 =	4323	120 =	1551
30 =	7117	80 =	2848	130 =	0
40 =	5748	90 =	8879	140 =	959
50 =	7333	100 =	3201	150 =	199

160 =	176	370 =	2271	580 =	4547
170 =	1404	380 =	1233	590 =	2049
180 =	6125	390 =	1039	600 =	201
190 =	1648	400 =	1560	610 =	0
200 =	1821	410 =	1313	620 =	3038
210 =	1818	420 =	775	630 =	2785
220 =	1868	430 =	1276	640 =	3068
230 =	1931	440 =	619	650 =	2825
240 =	2183	450 =	1320	660 =	201
250 =	4124	460 =	201	670 =	0
260 =	4114	470 =	0	680 =	1606
270 =	5248	480 =	2018	690 =	3069
280 =	3507	490 =	2034	700 =	3343
290 =	330	500 =	201	710 =	1626
300 =	0	510 =	0	720 =	176
310 =	304	520 =	5508	730 =	2940
320 =	1996	530 =	4735	740 =	1288
330 =	1704	540 =	4821	750 =	1649
340 =	2373	550 =	414		
350 =	201	560 =	1406	TOTAL =	171546
360 =	0	570 =	2935		